

Bachelor/Master Exam

First Name: _____

Last Name: _____

Immatriculation Number: _____

Course of Studies (please mark exactly one):

- Informatik Bachelor**
- Informatik Master**
- Mathematik Master**
- SSE Master**
- Other:** _____

	Maximal Points	Achieved Points
Exercise 1	8	
Exercise 2	10	
Exercise 3	8	
Exercise 4	10	
Exercise 5	9	
Exercise 6	7	
Exercise 7	8	
Total	60	
Grade	-	

Instructions:

- On every sheet please give your **first name**, **last name**, and **immatriculation number**.
- You must solve the exam **without** consulting any **extra documents** (e.g., course notes).
- Make sure your answers are readable. Do not use **red or green pens or pencils**.
- You can solve the exercises in **English** or **German**.
- Please answer the exercises on the **exercise sheets**. If needed, also use the back sides of the exercise sheets.
- Answers on extra sheets can only be accepted if they are clearly marked with your name, your immatriculation number, and the **exercise number**.
- **Cross out** text that should not be considered in the evaluation.
- Students that try to cheat **do not pass** the exam.
- At the end of the exam, please return **all sheets together with the exercise sheets**.

Exercise 1 (Theoretical Foundations):

(5 + 3 = 8 points)

Let $\varphi = p(0, s^2(0)) \wedge \forall X, Y \exists U (p(X, s(Y)) \rightarrow p(s^2(X), U)) \wedge \neg p(s^2(0), 0)$ and $\psi = \exists Z p(s^2(0), Z)$ be formulas over the signature (Σ, Δ) with $\Sigma = \Sigma_0 \cup \Sigma_1$, $\Sigma_0 = \{0\}$, $\Sigma_1 = \{s\}$, and $\Delta = \Delta_2 = \{p\}$.

a) Prove that $\{\varphi\} \models \psi$ by means of SLD resolution.

Hint: First transform the formula $\varphi \wedge \neg\psi$ into a clause set that is satisfiable iff $\varphi \wedge \neg\psi$ is satisfiable.

b) Explicitly give a Herbrand model of the formula φ (i.e., specify a carrier and a meaning for all function and predicate symbols). You do not have to provide a proof for your answer.

Hint:

- Here, $s^2(0)$ stands for $s(s(0))$ and $s^2(X)$ stands for the term $s(s(X))$.

Solution: _____

a)

$$\begin{aligned}
 \varphi \wedge \neg\psi &\Leftrightarrow p(0, s^2(0)) \wedge \forall X, Y \exists U (p(X, s(Y)) \rightarrow p(s^2(X), U)) \wedge \neg p(s^2(0), 0) \wedge \neg \exists Z p(s^2(0), Z) \\
 &\Leftrightarrow p(0, s^2(0)) \wedge \forall X, Y \exists U (\neg p(X, s(Y)) \vee p(s^2(X), U)) \wedge \neg p(s^2(0), 0) \wedge \neg \exists Z p(s^2(0), Z) \\
 &\Leftrightarrow p(0, s^2(0)) \wedge \forall X, Y \exists U (\neg p(X, s(Y)) \vee p(s^2(X), U)) \wedge \neg p(s^2(0), 0) \wedge \forall Z \neg p(s^2(0), Z) \\
 &\Leftrightarrow \forall X, Y, Z \exists U (p(0, s^2(0)) \wedge (\neg p(X, s(Y)) \vee p(s^2(X), U)) \wedge \neg p(s^2(0), 0) \wedge \neg p(s^2(0), Z))
 \end{aligned}$$


The above formula in prenex normal form is transformed into the following formula in Skolem normal form:

$$\forall X, Y, Z (p(0, s^2(0)) \wedge (\neg p(X, s(Y)) \vee p(s^2(X), a)) \wedge \neg p(s^2(0), 0) \wedge \neg p(s^2(0), Z))$$

Thus, the following clause set is satisfiable iff $\varphi \wedge \neg\psi$ is satisfiable:

$$\{p(0, s^2(0))\}, \{\neg p(X, s(Y)), p(s^2(X), a)\}, \{\neg p(s^2(0), 0)\}, \{\neg p(s^2(0), Z)\}.$$

We perform SLD resolution on this clause set to show $\{\varphi\} \models \psi$.



Hence, we have proven $\{\varphi\} \models \psi$.

□

b) We have $S \models \varphi$ for the Herbrand structure $S = (\mathcal{T}(\Sigma), \alpha)$ with $\alpha_0 = 0$, $\alpha_s(t) = s(t)$, and

$$\alpha_p = \{(s^i(0), s^j(0)) \mid i \geq 0, j \geq 1\}$$

Exercise 2 (Procedural Semantics, SLD tree):

(4 + 6 = 10 points)

Consider the following Prolog program \mathcal{P} which, for each pair of adjacent numbers in a list, swaps the numbers if the first number is greater than the second number.

```
bubble([], []).
bubble([X,Y|XS],[Y|YS]) :- X > Y, !, bubble([X|XS],YS).
bubble([X|XS],[X|YS]) :- bubble(XS,YS).
```

a) The program \mathcal{P}' results from \mathcal{P} by **removing the cut**. Consider the following query:

```
?- bubble([5,6,2],[5,X,Y]).
```

For the logic program \mathcal{P}' (i.e., **without the cut**), show a successful computation for the query above (i.e., a computation of the form $(G, \emptyset) \vdash_{\mathcal{P}'}^+ (\square, \sigma)$ where $G = \{\neg \text{bubble}([5,6,2],[5,X,Y])\}$). You may leave out the negations in the queries and you do not have to show how the substitutions operate on variables that only occur in (renamed) program clauses.

b) Please give a graphical representation of the SLD tree for the query

```
?- bubble([3,1,4],Res).
```

in the program \mathcal{P} (i.e., **with the cut**). For every part of a tree that is cut off by evaluating **!**, please indicate the cut by crossing out the edge that leads to the cut-off part. For the cut-off parts only indicate the first cut-off goal, but do not evaluate further. Please also indicate all answer substitutions.

Solution: _____


a)

```
({bubble([5,6,2],[5,X,Y])}, \emptyset)
\vdash_{\mathcal{P}'} ({bubble([6,2],[X,Y])}, \emptyset)
\vdash_{\mathcal{P}'} ({bubble([2],[Y])}, {X/6})
\vdash_{\mathcal{P}'} ({bubble([],[])}, {X/6,Y/2})
\vdash_{\mathcal{P}'} (\square, {X/6,Y/2})
```

Alternative solution:

```
({bubble([5,6,2],[5,X,Y])}, \emptyset)
\vdash_{\mathcal{P}'} ({bubble([6,2],[X,Y])}, \emptyset)
\vdash_{\mathcal{P}'} ({6 > 2, bubble([6],[Y])}, {X/2})
\vdash_{\mathcal{P}'} ({bubble([6],[Y])}, {X/2})
\vdash_{\mathcal{P}'} ({bubble([],[])}, {X/2,Y/6})
\vdash_{\mathcal{P}'} (\square, {X/2,Y/6})
```

b) SLD Tree



Answer substitution: {Res/[1,3,4]}

Exercise 3 (Definite Logic Programming):
(8 points)

We consider the following problem: Persons of different weights should be distributed onto boats with a capacity of 200 kg each. Here, we assume an infinite number of boats.

Implement a predicate `dist/2` in Prolog. If the first argument is a list of persons constructed using the function `person(w)` for a weight $w \in \mathbb{N}$, the predicate should return a list of boats as its second argument, where a boat is constructed using the function `boat(list)` for a list of persons `list`.

Assume that the persons are lined up and the boats are filled in a **greedy** way: The first boat is filled until the next person in line does not fit in (i.e., the sum of the weights of the people in the boat and the weight of the next person in the line exceeds 200 kg). Then, this person is the first to enter the next boat, etc. However, you should not use a new boat if the next person in line fits into the currently filled boat.

For example, the query `?- dist([person(140),person(70),person(70)], Boats)` should succeed with the only answer `Boats = [boat([person(140)]), boat([person(70),person(70)])]`. The query `?- dist([person(70),person(140),person(70)], Boats)` should succeed with the only answer `Boats = [boat([person(70)]), boat([person(140)]), boat([person(70)])]`.

You **may only use** the built-in predicates `is/2`, `</2`, `=</2`, the cut, and the usual arithmetic operators such as `+`, `*`, `-`, `/`. You may assume that no person in the line weighs more than 200 kg.

Solution: _____

```
weight([],0).
weight([person(W)|PS],R) :- weight(PS,R2), R is R2 + W.

dist([],[]).
dist([person(W)|PS],[boat([person(W)|G])|BS]) :- dist(PS,[boat(G)|BS]),
                                                    weight(G,A), A + W =< 200, !.
dist([person(W)|PS],[boat([person(W)])|BS]) :- dist(PS,BS).
```

Exercise 4 (Fixpoint Semantics):

(5 + 2 + 3 = 10 points)

Consider the following logic program \mathcal{P} over the signature (Σ, Δ) with $\Sigma = \{0, s\}$ and $\Delta = \{p\}$.

$p(0, X, X)$.
 $p(s(X), s(s(Y)), s(Z)) \text{ :- } p(X, Y, Z)$.

- a) For each $n \in \mathbb{N}$ explicitly give $\text{trans}_{\mathcal{P}}^n(\emptyset)$ in closed form, i.e., using a non-recursive definition.
- b) Compute the set $\text{lfp}(\text{trans}_{\mathcal{P}})$.
- c) Give $F[\mathcal{P}, \{\neg p(X, s(s(0)), Z)\}]$.

Solution: _____

Let G be the set of all ground terms, i.e., $G = \{s^i(0) \mid i \in \mathbb{N}\} = \mathcal{T}(\Sigma)$.

a)

$$\begin{aligned} \text{trans}_{\mathcal{P}}^0(\emptyset) &= \emptyset \\ \text{trans}_{\mathcal{P}}^1(\emptyset) &= \{p(0, t, t) \mid t \in G\} \\ \text{trans}_{\mathcal{P}}^2(\emptyset) &= \{p(s(0), s^2(t), s(t)) \mid t \in G\} \cup \text{trans}_{\mathcal{P}}^1(\emptyset) \\ \text{trans}_{\mathcal{P}}^3(\emptyset) &= \{p(s^2(0), s^4(t), s^2(t)) \mid t \in G\} \cup \text{trans}_{\mathcal{P}}^2(\emptyset) \\ \text{trans}_{\mathcal{P}}^4(\emptyset) &= \{p(s^3(0), s^6(t), s^3(t)) \mid t \in G\} \cup \text{trans}_{\mathcal{P}}^3(\emptyset) \\ &\vdots \\ \text{trans}_{\mathcal{P}}^n(\emptyset) &= \{p(s^i(0), s^{2i}(t), s^i(t)) \mid t \in G, 0 \leq i < n\} \end{aligned}$$

b)

$$\begin{aligned} \text{lfp}(\text{trans}_{\mathcal{P}}) &= \{p(s^i(0), s^{2i}(t), s^i(t)) \mid t \in G, i \geq 0\} \\ &= \{p(s^i(0), s^{2i+j}(0), s^{i+j}(0)) \mid t \in G, i \geq 0, j \geq 0\} \end{aligned}$$

c) $F[\mathcal{P}, \{\neg p(X, s(s(0)), Z)\}] = \{p(0, s^2(0), s^2(0)), p(s(0), s^2(0), s(0))\}$

_____.

Exercise 5 (Meta-Programming):

(9 points)

Implement a predicate `choose/2` which manipulates the term given as its first argument in such a way that for every subterm that is a compound of the form $f(t_1, \dots, t_n)$ with $n \geq 2$, all arguments except one are removed (i.e., every subterm of the form $f(t_1, \dots, t_n)$ is replaced by $f(t_i)$ for some $1 \leq i \leq n$). All other subterms should not be modified. For example, the query `?- choose(p(a,q(b,X)),R)` should succeed with the answers `R = p(a)`, `R = p(q(b))`, and `R = p(q(X))`.

You **may only use** the built-in predicates `var/1`, `atomic/1`, `compound/1`, and `=../2`.

Solution: _____

```

choose(Term,Term) :- var(Term).
choose(Term,Term) :- atomic(Term).
choose(Term,Res) :- compound(Term), Term =.. [Head|Args], mem(Arg,Args),
                    choose(Arg,ArgC), Res =.. [Head|[ArgC]].

mem(X,[X|_]).
mem(X,[_|XS]) :- mem(X,XS).

```


Exercise 6 (Universality):
(7 points)

Consider a function $f : \mathbb{N} \rightarrow \mathbb{N}$. The function $g : \mathbb{N} \times \mathbb{N} \rightarrow \mathbb{N}$ is defined as:

$$g(x, y) = k \text{ iff } f(x) + f(y) = f(k), \text{ } f(x), f(y), \text{ and } f(k) \text{ are defined, and} \\ \text{for all } 0 \leq k' < k, \text{ we have } f(x) + f(y) \neq f(k') \text{ and } f(k') \text{ is defined.}$$

As an example, consider the function $\hat{f} : \mathbb{N} \rightarrow \mathbb{N}$ with $\hat{f}(x) = |x - 4|$ (i.e., the absolute value of $x - 4$). The function $\hat{g} : \mathbb{N} \times \mathbb{N} \rightarrow \mathbb{N}$, constructed as described above, computes $\hat{g}(6, 3) = 1$. The reason is that for $x = 6, y = 3$, 1 is the smallest k such that $\hat{f}(x) + \hat{f}(y) = \hat{f}(k)$ and $\hat{f}(x) + \hat{f}(y) \neq \hat{f}(k')$ for all $0 \leq k' < k$. Indeed, $\hat{f}(6) + \hat{f}(3) = 2 + 1 = 3 = \hat{f}(1) = \hat{f}(k)$ and $\hat{f}(0) = 4 \neq 3$.

Consider a definite logic program \mathcal{P} which computes the function f using a predicate symbol $\underline{f} \in \Delta^2$:

$$f(x) = k \text{ iff } \mathcal{P} \models \underline{f}(x, k).$$

Here, numbers are represented by terms built from $0 \in \Sigma_0$ and $s \in \Sigma_1$ (i.e., $\underline{0} = 0, \underline{1} = s(0), \underline{2} = s(s(0)), \dots$).

Please extend the definite logic program \mathcal{P} such that it also computes the function g using the predicate symbol $\underline{g} \in \Delta^3$ (but without any built-in predicates):

$$g(x, y) = k \text{ iff } \mathcal{P} \models \underline{g}(x, y, k).$$

Hints:

- You may use the predicates `plus/3` and `ne/2` (for not equal):

```
plus(X,0,X).
plus(X,s(Y),s(Z)) :- plus(X,Y,Z).
```

```
ne(0,s(Y)).
ne(s(X),0).
ne(s(X),s(Y)) :- ne(X,Y).
```

Solution: _____

```
g(X1,X2,Z) :- f(X1,A1), f(X2,A2), plus(A1,A2,A), f'(0,Z,A).
f'(Z,Z,A) :- f(Z,A).
f'(Y,Z,A) :- f(Y,B), ne(A,B), f'(s(Y),Z,A).
```

Exercise 7 (Programming with CLP):
(8 points)

In this task, we use Prolog to solve simple addition equations where single digits may be replaced by variables. Here, a number is represented as the list of its digits. For example, the list $[2, 1, 7]$ represents the number 217.

Implement a Prolog predicate `add/3` such that `add(t_1, t_2, t_3)` is true if and only if t_1 , t_2 , and t_3 are lists of integers between 0 and 9 of the same length n and the sum of the numbers represented by t_1 and t_2 is the number represented by t_3 . For example, the query `?- add([X,4,6], [1,7,1], [2,Y,Z])` should succeed with the unique substitution $X = 0, Y = 1, Z = 7$.

The following line is already given:

```
:- use_module(library(clpfd)).
```

Hint:

- For any two natural numbers with n digits, their sum is a natural number with $n + 1$ digits where the most significant digit is either 0 or 1 (i.e., it corresponds to a carry bit). For example, $523 + 674 = 1197$ and $323 + 674 = 0997$. Thus, it could be helpful to implement an auxiliary predicate `add/4` where `add(t_1, t_2, t_3, c)` holds if the sum of the numbers represented by t_1 and t_2 is the number represented by $[c|t_3]$ where $c \in \{0, 1\}$.

Solution: _____

```
add(XS,YS,ZS) :- XS ins 0..9, YS ins 0..9, ZS ins 0..9, add(XS,YS,ZS,0).
```

```
add([],[],[],0).
```

```
add([X|XS],[Y|YS],[Z|ZS],NewCarry) :- add(XS,YS,ZS,OldCarry),
                                         Z + 10 * NewCarry #= X + Y + OldCarry.
```